Pràctica 3: Equacions i inequacions. Aplicacions.

En aquesta pràctica analitzarem algunes propietats de les funcions, com ara el càlcul d'arrels, màxims i mínims i intervals de creixement i decreixement, usant *Mathematica*. Utilitzarem les ordres per a agilitar el nostre treball i per a comprovar els resultats, tenint present els coneixements teòrics adquirits en l'assignatura.

1. Resolució d'equacions algebraiques

Al llarg de la pràctica plantejarem la resolució de tres tipus de problemes representats mitjançant equacions algebraiques:

- Càlcul d'arrels: $f(x) = 0$
- Punts de tall entre dues funcions: $f(x) = g(x)$
- Possibles extrems relatius de $f(x)$: $f'(x) = 0$

Tant per a trobar les arrels o els possibles extrems relatius d'una funció com per a calcular les abscisses dels punts de tall entre dues funcions, necessitarem utilitzar les ordres que posseeix *Mathematica* per a resoldre equacions: Solve, NSolve o FindRoot. Vegem, en primer lloc, la sintaxi i el funcionament d'aquestes tres ordres.

L'ordre **Solve** de *Mathematica* permet resoldre equacions, ja siga una única equació o un sistema d'equacions, utilitzant mètodes algebraics (sempre que això siga possible). És important assenyalar que les **equacions** s'introdueixen en *Mathematica* amb un doble signe d'igualtat:

expressió1 == expressió2

En el cas d'una única equació, l'ordre Solve té la sintaxi següent:

Solve[equació, variable]

Exemple. Resol l'equació $x^3 - 3x^2 + x - 3 = 0$.

```
In[1]:= Solve[x^3 - 3 x^2 + x - 3 == 0, x]
        resuelve
Out[1]= {{x → -i}, {x → i}, {x → 3}}
```

Si només busquem **solucions reals**, hi afegirem l'opció Reals com a argument:

Solve[equació, variable, Reals]

Exemple. Resol l'equació $e^x = 2$.

```
In[2]:= Solve[Exp[x] == 2, x, Reals]
        resue··· exponencial   números

Out[2]= {{x → Log[2]}}
```

Exemple. Resol l'equació $x^5 + 2x = 1$.

```
In[3]:= Solve[x^5 + 2 x == 1, x, Reals]
        resuelve                números

Out[3]= {{x → 0.486…}}
```

Veiem, en aquest últim exemple, que *Mathematica* no és capaç de trobar solucions exactes d'aquesta equació. Apareix una aproximació d'una solució real i, si premem sobre el símbol de l'arrel, observem que ha fet un càlcul en funció de l'ordre **Root**, però no és capaç d'obtenir una solució explícita. Això és natural per tractar-se d'una equació polinòmica de cinquè grau, per a la qual no hi ha una fórmula general de resolució.

En aquests casos podem usar l'ordre **NSolve**, que utilitza mètodes numèrics de resolució d'equacions per a trobar aproximacions decimals de les solucions. La sintaxi de l'ordre **NSolve** és la mateixa que la de l'ordre **Solve**:

NSolve[equació, variable]   o   NSolve[equació, variable, Reals]   (per a solucions reals)

Així, aplicant **NSolve** a l'exemple anterior, obtenim

```
In[4]:= NSolve[x^5 + 2 x == 1, x, Reals]
        resolvedor numérico      números

Out[4]= {{x → 0.486389}}
```

Si volem especificar un nombre de xifres significatives exactes, podem escriure

NSolve[equació, variable, WorkingPrecision $- > n$]

on n és el nombre de xifres:

```
In[5]:= NSolve[x^5 + 2 x == 1, x, Reals, WorkingPrecision → 10]
        resolvedor numérico      número·· precisión operativa

Out[5]= {{x → 0.4863890359}}
```

A vegades *Mathematica* ens diu que no pot aproximar les solucions de l'equació amb la instrucció **NSolve**. Es pot usar llavors una instrucció alternativa, **FindRoot**. Aquesta ordre aplica un mètode iteratiu per a aproximar numèricament **una única** solució de l'equació, per la qual cosa si el problema té diverses solucions caldrà executar l'ordre diverses vegades, i modificar-ne adequadament els arguments, per a trobar totes les solucions que busquem. La seua sintaxi admet dues variants:

FindRoot[equació, {variable, x_0}]   o   **FindRoot**[equació, {variable, a, b}]

En el primer cas, x_0 és un valor inicial del qual parteix el mètode iteratiu utilitzat per **FindRoot** per a aproximar l'arrel. Aquest punt ha de triar-se *prou* prop de la solució que volem aproximar; mentre que en el segon format, a y b són, respectivament, els extrems de l'interval [a, b] en què es troba la solució que busquem. Una representació gràfica prèvia[1] pot ser de gran utilitat a l'hora de seleccionar tant el punt inicial, x_0, com els extrems de l'interval [a, b].

També en l'ordre **FindRoot** podem especificar el nombre de xifres significatives exactes, afegint-hi l'opció WorkingPrecision $->n$.

Exemple. Observa gràficament que la funció $f(x) = e^x + \sin(3x)$ posseeix infinites arrels reals. Aplica l'ordre **FindRoot** per a aproximar l'arrel més pròxima a l'origen.

A partir d'una gràfica adequada podem observar que la funció posseeix infinites arrels. En efecte,

```
In[6]:= Plot[Exp[x] - Sin[3 x], {x, -10, 10}, PlotRange -> {-10, 20}]
        repr··· exponen··· seno                 rango de representación
```

Out[6]=



Comprova que no pots resoldre'l mitjançant l'ordre **NSolve**. Usarem, per tant, l'ordre

[1]Recorda que si et situes sobre la gràfica i tries l'opció *Obtenir coordenades* en el desplegable que apareix en prémer el botó dret del ratolí, el cursor es converteix en una creu i apareixen en pantalla les coordenades del punt sobre el qual es troba la creu.

FindRoot. Triant com a punt inicial del mètode $\mathtt{x}_0 = 0$, obtenim:

```
In[7]:= FindRoot[Exp[x] - Sin[3 x] == 0, {x, 0}]
        encuentra··· exponen··· seno

Out[7]= {x → -3.1558}
```

No obstant això, la solució que retorna el mètode no és la que busquem, ni tampoc la més pròxima al punt de partida. Observant la gràfica, veiem que -1 podria ser un punt inicial més adequat, ja que està més pròxim a l'arrel demanada. En aquest cas,

```
In[8]:= FindRoot[Exp[x] - Sin[3 x] == 0, {x, -1}]
        encuentra··· exponen··· seno

Out[8]= {x → -1.15413}
```

que retorna l'aproximació que ens demanaven.

Amb la finalitat de forçar el càlcul aproximat de l'arrel que busquem, convé indicar un interval en què està aquesta arrel (i només aquesta). En aquest cas, observem, a partir de la gràfica, que l'arrel demanada està en l'interval $[-2, -1]$ i podem obtenir-la mitjançant:

```
In[9]:= FindRoot[Exp[x] - Sin[3 x] == 0, {x, -2, -1}]
        encuentra··· exponen··· seno

Out[9]= {x → -1.15413}
```

A més, podem augmentar el nombre de decimals exactes de les aproximacions. Per exemple, per a aproximar la major de les arrels amb 20 dígits exactes, escrivim:

```
In[10]:= FindRoot[Exp[x] - Sin[3 x] == 0, {x, -2, -1}, WorkingPrecision → 20]
         encuentra··· exponen··· seno                  precisión operativa

Out[10]=
         {x → -1.1541326646068919041}
```

2. Resolució d'inequacions

Les inequacions es plantegen igual que les equacions però substituint el signe $==$ per $<, \leq, >, \geq$, segons corresponga. Per a resoldre una inequació del tipus

$$f(x) \leq g(x)$$

(o amb $\geq, <, >$ en lloc de $\leq$) podem emprar la instrucció **Reduce**, la sintaxi de la qual és

Reduce[inequació, variable]

Exemple. Resol la inequació $x^2 + 2x \geq 2$.

```
In[11]:= Reduce[x^2 + 2 x ≥ 2, x]
         reduce
Out[11]=
         x ≤ -1 - √3 || x ≥ -1 + √3
```

A l'hora d'interpretar els resultats hem de tenir en compte que el símbol `||` significa "o" (disjunció o unió de conjunts), mentre que `&&` denotaria "i" (conjunció o intersecció de conjunts). Per tant, la solució de la inequació anterior seria la unió dels intervals:

$$\left]-\infty, -1-\sqrt{3}\right] \cup \left[-1+\sqrt{3}, +\infty\right[.$$

A vegades, **Reduce** no resol la inequació si no s'indica el conjunt en què busquem les solucions. Per a això cal incloure un tercer argument, dominio:

Reduce[inequació, variable, dominio]

Normalment serà Reals, ja que buscarem solucions reals, encara que també podríem triar Integers o Complexes.

Exemple. Resol la inequació següent:

$$\operatorname{arctg}(x^3 + x) < 0$$

Observa que, si no s'especifica el domini, el programa retorna un missatge d'error:

```
In[12]:= Reduce[ArcTan[x^3 + x] < 0, x]
         reduce  arco tangente
         ••• Reduce: This system cannot be solved with the methods available to Reduce.
Out[12]=
         Reduce[ArcTan[x + x^3] < 0, x]
```

Aquest problema es pot evitar indicant que ens interessen només les solucions reals. Així,

```
In[13]:= Reduce[ArcTan[x^3 + x] < 0, x, Reals]
         reduce  arco tangente              números
Out[13]=
         x < 0
```

Per tant, la solució és l'interval $]-\infty, 0[$.

L'ordre **Reduce** també resol sistemes d'inequacions, que s'introdueixen entre claus com a primer paràmetre de l'ordre.

Exemple. Resol el sistema d'inequacions:

$$x^2 + 2x \geq 2, \;\; x - 4 \leq 0$$

Així

```
In[14]:= Reduce[{x^2 + 2 x ≥ 2, x - 4 ≤ 0}, x]
         reduce
Out[14]=
         x ≤ -1 - √3 || -1 + √3 ≤ x ≤ 4
```

Per la qual cosa la solució és

$$\left]-\infty, -1-\sqrt{3}\right] \cup \left[-1+\sqrt{3}, 4\right].$$

3. Càlcul de derivades. Aplicacions.

Per a calcular la derivada d'una funció $f(x)$, prèviament definida amb *Mathematica*, disposem de les ordres següents:

- **f´**[x] o **D**[f[x],x]: Calcula la derivada primera de f respecte de x, és a dir, $f'(x)$.
- **f´´**[x] o bé **D**[f[x],{x,2}]: Troba la derivada segona de f, això és $f''(x)$.
- **D**[f[x],{x,n}]: Calcula la derivada d'ordre n de f, és a dir, $f^{(n)}(x)$.

Exemple. Troba la derivada primera de $f(x) = x^3 - 4x^2 + x + 6$.

```
In[15]:= f[x_] := x^3 - 4 x^2 + x + 6

In[16]:= f'[x]

Out[16]=
         1 - 8 x + 3 x^2

In[17]:= D[f[x], x]
         deriva
Out[17]=
         1 - 8 x + 3 x^2
```

Si volem avaluar la derivada primera en $x = a$ podem fer-ho de dues maneres:

f´[a] o bien mediante **D**[f[x],x] $/.\ x \to a$

La mateixa regla se segueix per a derivades d'ordres superiors.

Exemple. Calcula $f'''(1)$, sent $f(x) = e^{x^2} + \cos(x)$.

```
In[18]:= f[x_] := Exp[x^2] + Cos[x]
                  exponencial  coseno

In[19]:= f'''[1]

Out[19]=
         20 e + Sin[1]

In[20]:= D[f[x], {x, 3}] /. x → 1
         deriva

Out[20]=
         20 e + Sin[1]
```

Observa, tanmateix, que no és correcte usar **D[f[a],{x,n}]** ja que, en aquest cas, calcula la derivada de $f(a)$ respecte de x que, en ser constant, sempre val 0. Comprova-ho.

En general, es poden estudiar algunes propietats geomètriques de les funcions analitzant-ne la derivada. Recorda els resultats teòrics següents:

Creixement i decreixement

Si $f'(x) > 0$ en un interval $I \Rightarrow f$ és estrictament creixent en I

Si $f'(x) < 0$ en un interval $I \Rightarrow f$ és estrictament decreixent en I

Càlcul d'extrems relatius

Si f és derivable i té un extrem relatiu en $x_0 \Rightarrow f'(x_0) = 0$

Per tant, els possibles extrems relatius d'una funció es troben resolent l'equació[2] $f'(x) = 0$. Una vegada calculats aquests valors, per a determinar si es tracta d'un màxim o un mínim, cal tenir en compte que en un mínim relatiu la corba és còncava i en un màxim relatiu la corba és convexa, és a dir,

Si $f'(x_0) = 0$ i $f''(x_0) > 0$, tenim un mínim en x_0

Si $f'(x_0) = 0$ i $f''(x_0) < 0$, tenim un màxim en x_0.

Ja que els possibles extrems relatius d'una funció derivable, $f(x)$, s'obtenen resolent l'equació $f'(x) = 0$ i els intervals de creixement i decreixement de $f(x)$ es poden trobar estudiant el signe de $f'(x)$, és a dir, resolent la inequació $f'(x) > 0$ (f creixent) o

[2] No sempre $f'(x) = 0$ implica l'existència d'un extrem relatiu

$f'(x) < 0$ (f decreixent), podem usar les ordres estudiades en els apartats anteriors amb *Mathematica* per a resoldre aquest tipus de problemes.

Exemple. Calcula els màxims i mínims relatius de $f(x) = x^4 + x^3 - 10x^2 + 8x$.

Comencem definint la funció amb *Mathematica*:

```
In[21]:= f[x_] := x^4 + x^3 - 10 x^2 + 8 x
```

Per a trobar els possibles extrems relatius resolem l'equació $f'(x) = 0$:

```
In[22]:= Solve[f'[x] == 0, x]
         resuelve
Out[22]=
         {{x → -2.79...}, {x → 0.448...}, {x → 1.60...}}

In[23]:= N[%]
         valor numérico
Out[23]=
         {{x → -2.79496}, {x → 0.448119}, {x → 1.59684}}
```

A partir d'una gràfica adequada observem que la funció posseeix un màxim (en $x = 0.448119$) i dos mínims relatius, que es corresponen amb els altres dos valors obtinguts. En efecte,

```
In[24]:= Plot[f[x], {x, -4, 4}]
         representación gráfica
Out[24]=
```



Alternativament, podríem haver classificat els extrems relatius avaluant el signe de la derivada segona de f en aquests valors.

Finalment, trobem el valor de les ordenades en els extrems relatius:

```
In[25]:= f[-2.79496]
Out[25]=
         -61.2871
```

```
In[26]:= f[0.448119]
Out[26]=
         1.70716
```

```
In[27]:= f[1.59684]
Out[27]=
         -2.1505
```

I podem concloure que $f(x)$ posseeix un màxim relatiu, $(0.448119,\ 1.70716)$, i dos mínims relatius, de coordenades $(-2.79496,\ -61.2871)$ i $(1.59684,\ -2.1505)$, respectivament.

El programa disposa de funcions específiques per a trobar els màxims i mínims d'una funció, $f(x)$, prèviament definida amb *Mathematica*:

- **FindArgMin**[f$[x], \{x, x_0\}$]: Troba un mínim de la funció f prop de x_0
- **FindArgMax**[f$[x], \{x, x_0\}$]: Troba un màxim de la funció f prop de x_0
- **FindMinimum**[f$[x], \{x, x_0\}$]: Retorna un parell de valors, on el segon és l'abscissa del punt en el qual la funció aconsegueix un mínim relatiu prop de x_0 i el primer és el valor de la funció en aquest mínim relatiu.
- **FindMaximum**[f$[x], \{x, x_0\}$]: Retorna un parell de valors, on el segon és l'abscissa del punt en el qual la funció aconsegueix un màxim relatiu prop de x_0 i el primer és el valor de la funció en aquest màxim relatiu.
- **Minimize**[f$[x], x$]: Retorna, si la funció té mínim absolut, un parell de valors, en què el segon és l'abscissa del punt on la funció aconsegueix aquest mínim absolut i el primer és el valor del mínim absolut.
- **Maximize**[f$[x], x$]: Retorna, si la funció té màxim absolut, un parell de valors, en què el segon és l'abscissa del punt on la funció aconsegueix aquest màxim absolut i el primer és el valor del màxim absolut.

Encara que aquestes ordres poden ser de gran utilitat per a trobar màxims o mínims, no sempre funcionen bé, per la qual cosa recomanem la seua aplicació només com a suport o confirmació dels resultats que obtens usant directament les definicions teòriques corresponents.

Per a finalitzar la pràctica, utilitzarem les ordres i procediments descrits en els apartats anteriors per a determinar algunes propietats d'una funció, amb ajuda de *Mathematica*.

Ejemplo. Donada la funció racional:

$$f(x) = \frac{3x^3 - x^2 - 6x + 3}{x^2 + x - 1}$$

- Representa-la gràficament i troba el seu domini.
- Calcula les seues arrels exactes i/o aproximades.
- Troba els punts de tall de $f(x)$ amb la recta $y = 3x$.
- Troba, usant el comando **Reduce**, els punts en els quals $f(x)$ es troba per davall de $3x$
- Determina les coordenades dels extrems relatius de $f(x)$ i indica els seus intervals de creixement i decreixement.

Comencem per definir $f(x)$ amb *Mathematica*:

```
In[28]:= f[x_] := (3 x^3 - x^2 - 6 x + 3) / (x^2 + x - 1)
```

A continuació, mostrarem la seua represenció gràfica:

```
In[29]:= Plot[f[x], {x, -5, 5}]
         representación gráfica
Out[29]=
```



Observa que es tracta d'una funció racional, per la qual cosa el seu domini seran tots els nombres reals excepte les arrels del denominador, que podem trobar mitjançant **Solve**:

```
In[30]:= Solve[x^2 + x - 1 == 0, x]
          resuelve
Out[30]=
{{x → 1/2 (-1 - √5)}, {x → 1/2 (-1 + √5)}}
```

Per tant, el domini de $f(x)$ és

$$\mathbb{R} \sim \left\{ -\frac{\sqrt{5}}{2} - \frac{1}{2}, \frac{\sqrt{5}}{2} - \frac{1}{2} \right\}$$

A més,

$$x = -\frac{\sqrt{5}}{2} - \frac{1}{2} \quad \text{i} \quad x = \frac{\sqrt{5}}{2} - \frac{1}{2}$$

són asímptotes verticals de $f(x)$, com s'aprecia en la gràfica.

Observem també que $f(x)$ té tres arrels reals, ja que la gràfica talla l'eix OX en tres punts (noteu que encara que només visualitzem la gràfica en l'interval $[-5,\ 5]$ no hi pot haver més talls atès que la funció del numerador és un polinomi de grau 3). Per a trobar les arrels exactes hem de resoldre l'equació $f(x) = 0$. Així

```
In[31]:= Solve[f[x] == 0, x]
          resuelve
Out[31]=
{{x → -1.48...}, {x → 0.527...}, {x → 1.28...}}
```

que no retorna el resultat exacte, a pesar que les equacions polinòmiques de tercer grau (numerador) sí que es poden resoldre de manera exacta. Per a trobar una aproximació numèrica de les arrels usarem **NSolve** (també pots aproximar el resultat anterior amb **N**[%]):

```
In[32]:= NSolve[f[x] == 0, x]
          resolvedor numérico
Out[32]=
{{x → -1.47785}, {x → 1.28432}, {x → 0.52686}}
```

Alternativament, podem aproximar les arrels mitjançant **FindRoot**, tenint en compte que hem d'executar-lo tantes vegades com arrels cerquem, ja que només retorna una aproximació cada vegada. Així, podem trobar una aproximació a l'arrel de $f(x)$, triant com a punt inicial $x_0 = 1$, mitjançant:

```
In[33]:= FindRoot[f[x] == 0, {x, 1}]
         encuentra raíz
Out[33]=
         {x → 1.28432}
```

Com ja hem comentat, en funcions amb més d'una arrel, no és possible saber quina solució proporcionarà el mètode quan s'usa la sintaxi de **FindRoot** amb x_0. En efecte, observa que fins i tot, encara que el punt triat estiga prop d'una arrel, pot aproximar una altra més allunyada:

```
In[34]:= FindRoot[f[x] == 0, {x, -0.5}]
         encuentra raíz
Out[34]=
         {x → 1.28432}
```

Amb la finalitat de forçar el càlcul aproximat de la solució que cerquem, convé indicar un interval en el qual està aquesta arrel (i només aquesta). Per a això, usarem el segon format del comando **FindRoot**, ajudant-nos prèviament d'una representació gràfica adequada per a seleccionar els extrems de l'interval que hem d'introduir. Utilitzant aquesta ordre podem aproximar les tres arrels de $f(x)$ mitjançant:

```
In[35]:= FindRoot[f[x] == 0, {x, -1.5, -1}]
         encuentra raíz
Out[35]=
         {x → -1.47785}

In[36]:= FindRoot[f[x] == 0, {x, 0, 0.57}]
         encuentra raíz
Out[36]=
         {x → 0.52686}

In[37]:= FindRoot[f[x] == 0, {x, 1, 2}]
         encuentra raíz
Out[37]=
         {x → 1.28432}
```

Respecte al següent apartat, les abscisses dels punts de tall entre la funció $f(x)$ i la recta $y = 3x$ s'obtenen resolent l'equació:

$$f(x) = 3x$$

Usant *Mathematica*:

```
In[38]:= Solve[f[x] == 3 x, x]
         resuelve
Out[38]=
```

$$\left\{\left\{x \to \frac{1}{8}\left(-3-\sqrt{57}\right)\right\}, \left\{x \to \frac{1}{8}\left(-3+\sqrt{57}\right)\right\}\right\}$$

Per a esbrinar en quines regions el gràfic de $y = f(x)$ es troba per davall de la recta $y = 3x$, n'hi ha prou amb utilitzar la seqüència

```
In[39]:= Reduce[f[x] < 3 x, x]
         reduce
Out[39]=
```

$$x < \frac{1}{2}\left(-1-\sqrt{5}\right) \;||\; \frac{1}{8}\left(-3-\sqrt{57}\right) < x < \frac{1}{8}\left(-3+\sqrt{57}\right) \;||\; x > \frac{1}{2}\left(-1+\sqrt{5}\right)$$

Per tant, la solució de la inequació $f(x) < 3x$ és la unió dels intervals:

$$\left]-\infty, -\frac{\sqrt{5}}{2}-\frac{1}{2}\right[\;\cup\; \left]-\frac{\sqrt{57}}{8}-\frac{3}{8}, \frac{\sqrt{57}}{8}-\frac{3}{8}\right[\;\cup\; \left]\frac{\sqrt{5}}{2}-\frac{1}{2}, +\infty\right[$$

Així, $f(x)$ es troba per davall de $3x$ entre els punts d'intersecció entre ambdues, $x = -\frac{\sqrt{57}}{8} - \frac{3}{8}$ i $x = \frac{\sqrt{57}}{8} - \frac{3}{8}$, a l'esquerra de la seua asímptota $x = -\frac{\sqrt{5}}{2} - \frac{1}{2}$ i a la dreta de la seua asímptota $x = \frac{\sqrt{5}}{2} - \frac{1}{2}$. Pots comprovar-ho mitjançant un gràfic adequat, representant totes dues funcions, la recta $y = 3x$ al costat de $y = f(x)$, superposades.

Com a aplicació del càlcul de derivades podem determinar les coordenades dels punts on la funció $f(x)$ aconsegueix el màxim i el mínim relatiu que s'aprecien en la gràfica. Ja que busquem solucions de l'equació $f'(x) = 0$, per a trobar les abscisses d'aquests punts n'hi haurà prou amb aplicar **Solve** per a resoldre l'equació corresponent:

```
In[40]:= Solve[f'[x] == 0, x]
         resuelve
Out[40]=
{{x → -1}, {x → -2.24...}, {x → 0.620... - 0.248... i}, {x → 0.620... + 0.248... i}}
```

que, després d'aproximar el resultat, resulta

```
In[41]:= N[%]
         valor numérico
Out[41]=
{{x → -1.}, {x → -2.24059}, {x → 0.620294 - 0.248087 i}, {x → 0.620294 + 0.248087 i}}
```

De les dues[3] solucions reals que obtenim

$$x = -1 \qquad \text{i} \qquad x \approx -2.24059$$

resulta evident que $x = -1$ és la que correspon[4] al mínim relatiu i l'altra al màxim relatiu. El mínim relatiu s'aconsegueix, doncs, en el punt de coordenades

$$(-1, f(-1)) = (-1, -5),$$

mentre que el màxim relatiu s'aconsegueix en (comproveu-ho)

$$(-2.24059, f(-2.24059)) \approx (-2.24059, -12.5427).$$

Finalment, podem trobar els intervals de creixement d'aquesta funció aplicant l'ordre **Reduce** per a resoldre la inequació

$$\mathtt{f}'(\mathtt{x}) > 0$$

```
In[42]:= Reduce[f'[x] > 0, x]
         reduce
Out[42]=
         x < -2.24… || -1 < x < 1/2 (-1 + √5) || x > 1/2 (-1 + √5)
```

i obtindràs, després d'aproximar l'expressió,

```
In[43]:= N[%]
         valor numérico
Out[43]=
         x < -2.24059 || -1. < x < 0.618034 || x > 0.618034
```

Comprova el resultat observant la gràfica. Anàlogament, podries trobar els intervals de decreixement resolent la inequació $f'(x) < 0$.

[3] $f'(x) = 0$ té quatre solucions però les dues complexes no tenen interpretació gràfica.
[4] La prova analítica d'aquest fet és que $f'(-1) = 0$ i $f''(-1) > 0$.